

Word Review for Presentation about the %Caption macro

1. Directions for inserting a file into a Word document.
 - a. Put cursor in Word document where you want to insert the file.
 - b. Click Insert from the menu bar and scroll down to File and click again.
 - c. Pop up box appears: Find the file you want to insert and click it ONCE.
 - d. Move the cursor to the bottom right of the pop up box and click the Insert ▼ and then click on Insert as Link.
2. Directions for selecting the whole Word document.
 - a. Click Edit from the menu bar and scroll down to Select All.
 - b. OR hit Ctrl+A
3. Directions for updating field codes once selected.
 - a. Right click on the highlighted area, which will be the whole document if you followed #2 above, and then click !Update Field from the pop up box.
 - b. Once the area you want is highlighted, hit F9.
4. Directions for inserting a bookmark into the text.
 - a. Put cursor in Word document where you want to insert a bookmark.
 - b. Click Insert from the menu bar and scroll down to Reference. Then slide the cursor to the right and click on Cross-reference.
 - c. A pop up box appears and make Reference type = Bookmark.
 - d. Click on a bookmark name you made in the %caption macro in the “For which bookmark” box and then click the Insert button. In my example, you would look for TABGENDER in the “For which bookmark” box.
 - e. You will see the table number appear in your document like this: Table 1, or Figure 1.1, or Table 2a, or whatever number it is.
 - f. If the table numbers change, these bookmark references will also change with them. So, if Table 1 became Table 3 and there was a bookmark for this table, then after updating the field the bookmark would also change to Table 3.

```

/*****/
/* PROGRAM:    caption.sas
/* PURPOSE:    Puts an automatic caption in an RTF file
/* OUTPUT:
/* AUTHOR:     Heather Ribaud
/* CREATION DATE: 02/01/07
/* NOTES:      swtch can take the values n (insert the Next sequence number),
                c (use the Current sequence number, cnum likely to be a letter, a or b)
                s (index to heading number, snum should equal the heading level, so if the heading is 3.1.2, then snum=3),
                r (Reset sequence number to value given in rnum),
/*****/

%macro caption(captiontext=,tnum=1,type=Table,justify=left,bkmrk=,swtch=n,snum=,cnum=,rnum=) ;
  %if %lowercase(&swtch)=n %then
    title&tnum j=&justify "{{\*bkmkstart &bkmrk}{&type \field{\*fldinst{SEQ &type \* ARABIC \n }}}{\*bkmkend &bkmrk}: &captiontext}" ;;

  %if %lowercase(&swtch)=c %then
    title&tnum j=&justify "{{\*bkmkstart &bkmrk}{&type \field{\*fldinst{SEQ &type \* ARABIC \c }}}&cnum{\*bkmkend &bkmrk}: &captiontext}" ;;

  %if %lowercase(&swtch)=r %then
    title&tnum j=&justify "{{\*bkmkstart &bkmrk}{&type \field{\*fldinst{SEQ &type \* ARABIC \r &rnum }}}{\*bkmkend &bkmrk}: &captiontext}" ;;

  %if %lowercase(&swtch)=s %then
    title&tnum j=&justify "{{\*bkmkstart &bkmrk}{&type {\field{\*fldinst{STYLEREf &snum \s}}.\field{\*fldinst{SEQ &type \* ARABIC \s
&snum}}}}{\*bkmkend &bkmrk}: &caption
text}" ;;
%mend ;

```

```

/* ****

```

EXAMPLES FOR USING Caption macro

**** In its most basic form ;**

```
%caption(captiontext=Basic use,bkmrk=basic) ;
```

**** Gives:**

Table 1: Basic use

**** When you need sub-titles ;**

```
%caption(captiontext=Main heading,bkmrk=main) ;
```

```
%caption(captiontext=Part one,bkmrk=mainA,swtch=c,cnum=a) ;
```

```
<Procedure>
```

```
title1 ;
```

```
%caption(captiontext=Part two,bkmrk=mainB,swtch=c,cnum=b) ;
```

```
** Gives:
```

```
Table 2: Main heading
```

```
Table 2a: Part one
```

```
<Procedure>
```

```
Table 2b: Part two
```

```
** Matching to section heading numbers;
```

```
** Headings must already be in the document that the tables are inserted into to see the coordinating section numbers;
```

```
%caption(captiontext=My Awesome Table,bkmrk=awesome,swtch=s,snum=2);
```

```
< Procedure>
```

```
** Gives:
```

```
2 Results
```

```
2.1 Africa Results
```

```
Table 2.1: My Awesome Table
```

```
<Procedure>
```

NOTE: This macro can be altered so that instead of 3 parameters (snum, cnum, and rnum) there is only 1 (maybe call it swnum), which could be helpful if these become parameters to a macro that may produce many tables based on the values in these parameters, which we have at CBAR.

BONUS MACROS – %NO_SECTION and %WORDSTYLE:

```

/*****
/* PROGRAM:    no_section.sas
/* PURPOSE:    Remove section breaks from a RTF output file,
/*             except for first one which appears necessary for formatting,
/*             like orientation=landscape.
/* INPUT:
/* OUTPUT:
/* AUTHOR:      Obtained from SAS (Wayne Hester) Feb 2, 2007
/*             through an email to Brendan Bartley
/* CREATION DATE: 02/07/07
/* NOTES:       Use filename statement as reference for
/*             the macro variable, SOURCE
/*             Use this after the ODS rtf close;
/*             EXAMPLE BELOW
/*             filename myreport '/home/bbartley/myreport.rtf';
/*             %include '/home/prinf/programs/macros/no_section.sas';
/*             CODE INCLUDING ODS RTF statements finishing with
/*             ODS RTF close;
/*             %no_section(SOURCE=myreport);
/* MODIFICATIONS: Changed the code to keep the first section allowing
/*             orientation=landscape to work and fix other problems
/*             with some tables - BB, 1/22/2008
*****/

```

```

*****;
* Notice that the title and footnote will be the last one set
* since there is no section boundary ;

```

```

%macro no_section(source=xxx);
/*
    Reading and writing to the same file at the same time can be trouble
    depending on where you are.
*/

```

```

data temp ;
length line $400;
infile &source length=lg lrecl=1000 end=eof;
input @1 line $varying400. lg;

```

```

data _null_;
  set temp ; retain flag 0 ;
  file &source;

  /*
    Trap for section data line only.
    Don't insert a space line for the first occurrence,
    only between tables.
  */

  /*
    Keep the first section information which seems to
    define the characteristics of the document.
  */

  if (index(line,"\sect") > 0) then do ;
    if ( flag ) then
      line = "{\pard\par}" ;
    else do ;
      flag = 1 ;
    end ;
  end ;

  put line ;

run;

%mend no_section;

```

HERE IS AN EXAMPLE HOW TO USE IT:

```

filename aname "/home/bbartley/mystudy/bfile.rtf";
ods rtf file = aname <other ods options, probably including BODYTITLE>;

```

<CODE FOR TABLES OR FIGURES IN YOUR FILE>

```

ods rtf close;
%no_section(source=aname);

```

```

/*****/
/* PROGRAM:      /home/ribaudo/sasmacros/wordstyle.sas
/* PURPOSE:      To change the default wordstyle name in a given RTF file
/* INPUT:        source=   name of RTF file to change
/*                wordstyle= name of style in Word to be used
/* OUTPUT:       RTF of same name with default style changed (defaults to RTFOutput - a style defined in the SDAC Word shell)
/*                It may be helpful to create your own Word shell to define your own preferred style for tables, figures, and graphs.
/* AUTHOR:       Heather Ribaud
/* CREATION DATE: 03/20/07
/* NOTES:
/* MODIFICATIONS:
/*  Chg #1:   Date: 7/30/07  By: Ray Griner
/*  Moved from WORDStyle.sas to wordstyle.sas
/*  Filename needs to be lowercase for autocall to work
/*****/

```

```
%macro wordstyle(source=,wordstyle=RTFOutput) ;
```

```
data temp ;
```

```
length line $400;
```

```
infile &source length=lg lrecl=1000 end=eof;
```

```
input @1 line $varying400. lg;
```

```
data _null_;
```

```
set temp ; retain flag 0 ;
```

```
file &source lrecl=500 ;
```

```
if index(line,"Normal;}")>0 then line=tranwrd(line,"Normal;}", "&wordstyle;");
```

```
put line ;
```

```
run ;
```

```
%mend ;
```

```
*****
```

This macro creates a style name for your tables, so that it is not the same as the “Normal” one used in your report.

Why would you want a different style name for your tables versus the style name for the text in your report?

Let’s say your text is double spaced and in Arial font, but you want your tables to be single spaced and use

Times New Roman font. This would allow you to do so, if you’ve created a style that accommodates both (“Normal” for your text, “YourStyleName” for your tables). The details of the style may be created after it is named.

Look in the Word documentation to see how this may be done. It would be used in SAS like the %no_section macro:

```
filename aname “/home/bbartley/mystudy/bfile.rtf”;
```

```
ods rtf file = aname <other ods options, probably including BODYTITLE>;
```

```
<CODE FOR TABLES OR FIGURES IN YOUR FILE, PROBABLY INCLUDING MANY %CAPTION MACRO CALLS>
```

```
ods rtf close;  
%wordstyle(source=aname,wordstyle=mystylename);
```

ALL THREE MACROS COULD BE USED TOGETHER TO CREATE A NICE DOCUMENT TO INSERT INTO YOUR REPORT.

```
filename aname "/home/bbartley/mystudy/bfile.rtf";  
ods rtf file = aname <other ods options, probably including BODYTITLE>;
```

```
<CODE FOR TABLES OR FIGURES IN YOUR FILE, PROBABLY INCLUDING MANY %CAPTION MACRO CALLS >
```

```
ods rtf close;  
%no_section(source=aname);  
%wordstyle(source=aname,wordstyle=mystylename);
```

Hopefully, there will be a NESUG paper properly documenting these macros more clearly, but for those adventurous enough to try new things, I figured I'd print these extra two macros that we've found very helpful here at CBAR. My email is bbartley@sdac.harvard.edu if you have questions.